What is ASP?

· ASP stands for Active Server Pages 

· ASP is a program that runs inside IIS 

· IIS stands for Internet Information Services 

· IIS comes as a free component with Windows 2000 

· IIS is also a part of the Windows NT 4.0 Option Pack 

· The Option Pack can be downloaded from Microsoft 

· PWS is a smaller - but fully functional - version of IIS 

· PWS can be found on your Windows 95/98 CD 

ASP Compatibility

· ASP is a Microsoft Technology 

· To run IIS you must have Windows NT 4.0 or later 

· To run PWS you must have Windows 95 or later 

· Chili ASP is a technology that runs ASP without Windows OS 

· InstantASP is another technology that runs ASP without Windows 

What is an ASP File?

· An ASP file is just the same as an HTML file 

· An ASP file can contain text, HTML, XML, and scripts 

· Scripts in an ASP file are executed on the server 

· An ASP file has the file extension ".asp" 

How Does ASP Differ from HTML?

· When a browser requests an HTML file, the server returns the file 

· When a browser requests an ASP file, IIS passes the request to the ASP engine. The ASP engine reads the ASP file, line by line, and executes the scripts in the file. Finally, the ASP file is returned to the browser as plain HTML 



What can ASP do for you?

· Dynamically edit, change or add any content of a Web page 

· Respond to user queries or data submitted from HTML forms 

· Access any data or databases and return the results to a browser 

· Customize a Web page to make it more useful for individual users 

· The advantages of using ASP instead of CGI and Perl, are those of simplicity and speed 

· Provides security since your ASP code can not be viewed from the browser 

· Since ASP files are returned as plain HTML, they can be viewed in any browser 

· Clever ASP programming can minimize the network traffic 

Run ASP on Your PC
You can run ASP on your own PC without an external server. To do that, you must install Microsoft's Personal Web Server (PWS) or Internet Information Services (IIS) on your PC.



How to Run ASP on your own PC

You can run ASP on your own PC without an external server. To do that, you must install Microsoft's Personal Web Server (PWS) or Internet Information Services (IIS) on your PC.

If you are serious about using ASP, you should have at least Windows 98, Second Edition. 
If you are really serious about using ASP, you should go for Windows 2000.


How to install PWS and run ASP on Windows 95

Personal Web Server (PWS) is not shipped with Windows 95 !!

To run ASP on Windows 95, you will have to download "Windows NT 4.0 Option Pack" from Microsoft.



How to install PWS and run ASP on Windows NT

Personal Web Server (PWS) is not shipped with Windows NT !!

To run ASP on Windows NT, you will have to download "Windows NT 4.0 Option Pack" from Microsoft.



How to install PWS and run ASP on Windows 98

1. Open the Add-ons folder on your Windows98 CD, find the PWS folder and run the setup.exe file. 

2. An Inetpub folder will be created on your hard drive. Open it and find the wwwroot folder. 

3. Create a new folder, like "MyWeb", under wwwroot. 

4. Use a text editor to write some ASP code, save the file as "test1.asp" in the "MyWeb" folder. 

5. Make sure your Web server is running - The installation program has added a new icon on your task bar (this is the PWS symbol). Click on the icon and press the Start button in the window that appears. 

6. Open your browser and type in "http://localhost/MyWeb/test1.asp", to view your first ASP page. 



How to install PWS and run ASP on Windows ME

Personal Web Server (PWS) is not included with Windows Me !!



How to install IIS and run ASP on Windows 2000

1. From your Start Button, go to Settings, and Control Panel 

2. In the Control Panel window select Add/Remove Programs 

3. In the Add/Remove window select Add/Remove Windows Components 

4. In the Wizard window check Internet Information Services, click OK 

5. An Inetpub folder will be created on your hard drive 

6. Open the Inetpub folder, and find a folder named wwwroot 

7. Create a new folder, like "MyWeb", under wwwroot. 

8. Use a text editor to write some ASP code, save the file as "test1.asp" in the "MyWeb" folder 

9. Make sure your Web server is running - The installation program has added a new icon on your task bar (this is the IIS symbol). Click on the icon and press the Start button in the window that appears. 

10. Open your browser and type in "http://localhost/MyWeb/test1.asp", to view your first ASP page 



How to install IIS and run ASP on Windows XP Professional

Note: You cannot run ASP on Windows XP Home Edition.

1. Insert the Windows XP Professional CD-Rom into your CD-Rom Drive 

2. From your Start Button, go to Settings, and Control Panel 

3. In the Control Panel window select Add/Remove Programs 

4. In the Add/Remove window select Add/Remove Windows Components 

5. In the Wizard window check Internet Information Services, click OK 

6. An Inetpub folder will be created on your hard drive 

7. Open the Inetpub folder, and find a folder named wwwroot 

8. Create a new folder, like "MyWeb", under wwwroot. 

9. Use a text editor to write some ASP code, save the file as "test1.asp" in the "MyWeb" folder 

10. Make sure your Web server is running - The installation program has added a new icon on your task bar (this is the IIS symbol). Click on the icon and press the Start button in the window that appears. 

11. Open your browser and type in "http://localhost/MyWeb/test1.asp", to view your first ASP page 

ASP Syntax
You cannot view the ASP source code by selecting "View source" in a browser, you will only see the output from the ASP file, which is plain HTML. This is because the scripts are executed on the server before the result is sent back to the browser.

In our ASP tutorial, every example displays the hidden ASP source code. This will make it easier for you to understand how it works.



Examples 

How to write some text with ASP. 

<html>
<body>

<%
response.write("Hello World!")
%>

</body>
</html>
How to format the text with HTML tags. 

<html>
<body>
<%
response.write("<h2>You can use HTML tags to format the text!</h2>")
%>

<%
response.write("<p style='color:#0000ff'>This text is styled with the style attribute!</p>")
%>
</body>
</html>


The Basic Syntax Rule 

An ASP file normally contains HTML tags, just like an HTML file. However, an ASP file can also contain server scripts, surrounded by the delimiters <% and %>. Server scripts are executed on the server, and can contain any expressions, statements, procedures, or operators valid for the scripting language you prefer to use. 



The Response Object 

The Write method of the ASP Response Object is used to send content to the browser. For example, the following statement sends the text "Hello World" to the browser:

	<%

response.write("Hello World!")

%>




VBScript

You may use different scripting languages in ASP files. However, the default scripting language is VBScript:

	<html>

<body>

<%

response.write("Hello World!")

%>
</body>

</html>


The example above writes "Hello World!" into the body of the document. 



JavaScript 

To set JavaScript as the default scripting language for a particular page you must insert a language specification at the top of the page:

	<%@ language="javascript"%>

<html>

<body>

<%

Response.Write("Hello World!")

%>
</body>

</html>


Note: Unlike VBScript - JavaScript is case sensitive. You will have to write your ASP code with uppercase letters and lowercase letters when the language requires it. 



Other Scripting Languages 

ASP is shipped with VBScript and JScript (Microsoft's implementation of JavaScript). If you want to script in another language, like PERL, REXX, or Python, you will have to install script engines for them.

Important: Because the scripts are executed on the server, the browser that displays the ASP file does not need to support scripting at all!

ASP Variables
A variable is used to store information. 

If the variable is declared outside a procedure it can be changed by any script in the ASP file. If the variable is declared inside a procedure, it is created and destroyed every time the procedure is executed. 



Examples

Declare a variable
Variables are used to store information. This example demonstrates how to declare a variable, assign a value to it, and use the value in a text.

<html>
<body>

<%
dim name
name="Donald Duck"
response.write("My name is: " & name)
%>

</body>
</html>
Declare an array
Arrays are used to store a series of related data items. This example demonstrates how to declare an array that stores names.

<html>
<body>

<%
Dim famname(6),i
famname(1) = "Jan Egil"
famname(2) = "Tove"
famname(3) = "Hege"
famname(4) = "Stale"
famname(5) = "Kai Jim"
famname(6) = "Borge"

For i = 1 to 6
      response.write(famname(i) & "<br />")
Next
%>

</body>
</html>
Loop through the HTML headers
How to loop through the six headers in HTML.

<html>
<body>

<%
dim i 
for i=1 to 6
   response.write("<h" & i & ">Header " & i & "</h" & i & ">")
next
%>

</body>
</html>
Time-based greeting using VBScript
This example will display a different message to the user depending on the time on the server.

<html>
<body>
<%
dim h
h=hour(now())

response.write("<p>" & now())
response.write(" (Norwegian Time) </p>")
If h<12 then
   response.write("Good Morning!")
else
   response.write("Good day!")
end if
%>
</body>
</html>
Time-based greeting using JavaScript
This example is the same as the one above, but the syntax is different.

<%@ language="javascript" %>
<html>
<body>
<%
var d=new Date()
var h=d.getHours()

Response.Write("<p>")
Response.Write(d + " (Norwegian Time)")
Response.Write("</p>")
if (h<12)
   {
   Response.Write("Good Morning!")
   }
else
   {
   Response.Write("Good day!")
   }
%>
</body>
</html>


Lifetime of Variables 

A variable declared outside a procedure can be accessed and changed by any script in the ASP file. 

A variable declared inside a procedure is created and destroyed every time the procedure is executed. No scripts outside the procedure can access or change the variable. 

To declare variables accessible to more than one ASP file, declare them as session variables or application variables.

Session Variables
Session variables are used to store information about ONE single user, and are available to all pages in one application. Typically information stored in session variables are name, id, and preferences.

Application Variables 
Application variables are also available to all pages in one application. Application variables are used to store information about ALL users in a specific application.

ASP Procedures

In ASP you can call a JavaScript procedure from a VBScript and vice versa. 



Examples 

Call a procedure using VBScript
How to call a VBScript procedure from ASP.

<html>

<head>
<%
sub vbproc(num1,num2)
response.write(num1*num2)
end sub
%>
</head>

<body>
<p>
You can call a procedure like this:
</p>
<p>
Result: <%call vbproc(3,4)%>
</p>
<p>
Or, like this:
</p>
<p>
Result: <%vbproc 3,4%>
</p>
</body>

</html>
Call a procedure using JavaScript
How to call a JavaScript procedure from ASP.

<%@ language="javascript" %>
<html>
<head>
<%
function jsproc(num1,num2)
{
Response.Write(num1*num2)
}
%>
</head>

<body>
<p>
Result: <%jsproc(3,4)%>
</p>
</body>

</html>
Call procedures using VBScript
How to call both a JavaScript procedure and a VBScript procedure in an ASP file.

<html>
<head>
<%
sub vbproc(num1,num2)
Response.Write(num1*num2)
end sub
%>
<script  language="javascript" runat="server">
function jsproc(num1,num2)
{
Response.Write(num1*num2)
}
</script>
</head>

<body>
<p>Result: <%call vbproc(3,4)%></p>
<p>Result: <%call jsproc(3,4)%></p>
</body>

</html>


Procedures 

The ASP source code can contain procedures and functions: 

	<html>

<head>

<%

sub vbproc(num1,num2)

response.write(num1*num2)

end sub

%>

</head>

<body>

<p>Result: <%call vbproc(3,4)%></p>

</body>

</html>


Insert the <%@ language="language" %> line above the <html> tag to write procedures or functions in another scripting language than default: 

	<%@ language="javascript" %>

<html>

<head>

<%

function jsproc(num1,num2)

{

Response.Write(num1*num2)

}

%>

</head>

<body>

<p>Result: <%jsproc(3,4)%></p>

</body></html>




Differences Between VBScript and JavaScript

When calling a VBScript or a JavaScript procedure from an ASP file written in VBScript, you can use the "call" keyword followed by the procedure name. If a procedure requires parameters, the parameter list must be enclosed in parentheses when using the "call" keyword. If you omit the "call" keyword, the parameter list must not be enclosed in parentheses. If the procedure has no parameters, the parentheses are optional. 

When calling a JavaScript or a VBScript procedure from an ASP file written in JavaScript, always use parentheses after the procedure name.

ASP Forms and User Input

The Request.QueryString and Request.Form commands may be used to retrieve information from forms, like user input.



Examples

A form with method="get"
How to interact with the user, with the Request.QueryString command.

<html>
<body>
<form action="demo_reqquery.asp" method="get">
Your name: <input type="text" name="fname" size="20">
<input type="submit" value="Submit">
</form>
<%
dim fname
fname=Request.QueryString("fname")
If fname<>"" Then
      Response.Write("Hello " & fname & "!<br />")
      Response.Write("How are you today?")
End If
%>
</body>
</html>
A form with method="post"
How to interact with the user, with the Request.Form command.

<html>
<body>
<form action="demo_simpleform.asp" method="post">
Your name: <input type="text" name="fname" size="20">
<input type="submit" value="Submit">
</form>
<%
dim fname
fname=Request.Form("fname")
If fname<>"" Then
      Response.Write("Hello " & fname & "!<br />")
      Response.Write("How are you today?")
End If
%>
</body>
</html>
A form with radio buttons
How to interact with the user, through radio buttons, with the Request.Form command.

<html>
<%
dim cars
cars=Request.Form("cars")
%>
<body>
<form action="demo_radiob.asp" method="post">
<p>Please select your favorite car:</p>

<input type="radio" name="cars"
<%if cars="Volvo" then Response.Write("checked")%>
value="Volvo">Volvo</input>
<br />
<input type="radio" name="cars"
<%if cars="Saab" then Response.Write("checked")%>
value="Saab">Saab</input>
<br />
<input type="radio" name="cars"
<%if cars="BMW" then Response.Write("checked")%>
value="BMW">BMW</input>
<br /><br />
<input type="submit" value="Submit" />
</form>
<%
if cars<>"" then
   Response.Write("<p>Your favorite car is: " & cars & "</p>")
end if
%>
</body>
</html>


User Input 

The Request object may be used to retrieve user information from forms:

	<form method="get" action="simpleform.asp">

First Name: <input type="text" name="fname">

<br />

Last Name: <input type="text" name="lname">

<br /><br />

<input type="submit" value="Submit">

</form>


User input can be retrieved in two ways: With Request.QueryString or Request.Form. 



Request.QueryString

The Request.QueryString command is used to collect values in a form with method="get". Information sent from a form with the GET method is visible to everyone (it will be displayed in the browser's address bar) and has limits on the amount of information to send.

If a user typed "Bill" and "Gates" in the form example above, the URL sent to the server would look like this:

	http://www.w3schools.com/simpleform.asp?fname=Bill&lname=Gates


Assume that the ASP file "simpleform.asp" contains the following script:

	<body>

Welcome

<%

response.write(request.querystring("fname"))

response.write(" " & request.querystring("lname"))

%>

</body>


The browser will display the following in the body of the document:

	Welcome Bill Gates




Request.Form 

The Request.Form command is used to collect values in a form with method="post". Information sent from a form with the POST method is invisible to others and has no limits on the amount of information to send.

If a user typed "Bill" and "Gates" in the form example above, the URL sent to the server would look like this: 

	http://www.w3schools.com/simpleform.asp


Assume that the ASP file "simpleform.asp" contains the following script:

	<body>

Welcome

<%

response.write(request.form("fname"))

response.write(" " & request.form("lname"))

%>

</body>


The browser will display the following in the body of the document:

	Welcome Bill Gates




Form Validation

User input should be validated on the browser whenever possible (by client scripts). Browser validation is faster and you reduce the server load.

You should consider using server validation if the user input will be inserted into a database. A good way to validate a form on the server is to post the form to itself, instead of jumping to a different page. The user will then get the error messages on the same page as the form. This makes it easier to discover the error.

ASP Cookies
A cookie is often used to identify a user.



Examples 

Welcome cookie
How to create a Welcome cookie.

<%
dim numvisits
response.cookies("NumVisits").Expires=date+365 
numvisits=request.cookies("NumVisits")

if numvisits="" then
   response.cookies("NumVisits")=1
   response.write("Welcome! This is the first time you are visiting this Web page.")
else
   response.cookies("NumVisits")=numvisits+1
   response.write("You have visited this ")
   response.write("Web page " & numvisits)
   if numvisits=1 then
     response.write " time before!"
   else
     response.write " times before!"
   end if
end if
%>
<html>
<body>
</body>
</html>


What is a Cookie? 

A cookie is often used to identify a user. A cookie is a small file that the server embeds on the user's computer. Each time the same computer requests for a page with a browser, it will send the cookie too. With ASP, you can both create and retrieve cookie values.



How to Create a Cookie

The "Response.Cookies" command is used to create cookies.

Note: The Response.Cookies command must appear BEFORE the <html> tag. 

In the example below, we will create a cookie named "firstname" and assign the value "Alex" to it:

	<%

Response.Cookies("firstname")="Alex"

%>


It is also possible to assign properties to a cookie, like setting a date when the cookie should expire: 

	<%

Response.Cookies("firstname")="Alex" 

Response.Cookies("firstname").Expires=#May 10,2002#

%>




How to Retrieve a Cookie Value

The "Request.Cookies" command is used to retrieve a cookie value. 

In the example below, we retrieve the value of the cookie named "firstname" and display it on a page:

	<%

fname=Request.Cookies("firstname")

response.write("Firstname=" & fname)

%>



Output: 

Firstname=Alex



A Cookie with Keys 

If a  cookie contains a collection of multiple values, we say that the cookie has Keys. 

In the example below, we will create a cookie collection named "user". The "user" cookie has Keys that contains information about a user:

	<%

Response.Cookies("user")("firstname")="John"

Response.Cookies("user")("lastname")="Smith"

Response.Cookies("user")("country")="Norway"

Response.Cookies("user")("age")="25"

%>




Read all Cookies

Look at the following code:

	<%

Response.Cookies("firstname")="Alex"

Response.Cookies("user")("firstname")="John"

Response.Cookies("user")("lastname")="Smith"

Response.Cookies("user")("country")="Norway"

Response.Cookies("user")("age")="25"

%>


Assume that your server has sent all the cookies above to a user.

Now we want to read all the cookies sent to a user. The example below shows how to do it (note that the code below checks if a cookie has Keys with the HasKeys property):

	<html>

<body>

<%

dim x,y

for each x in Request.Cookies

  response.write("<p>")

  if Request.Cookies(x).HasKeys then

    for each y in Request.Cookies(x)

      response.write(x & ":" & y & "=" & Request.Cookies(x)(y))

      response.write("<br />")

    next

  else

    Response.Write(x & "=" & Request.Cookies(x) & "<br />")

  end if

  response.write "</p>"

next

%>

</body>

</html>


Output:
firstname=Alex

user:firstname=John
user:lastname=Smith
user:country=Norway
user:age=25



What if a Browser Does NOT Support Cookies?

If your application deals with browsers that do not support cookies, you will have to use other methods to pass information from one page to another in your application. There are two ways of doing this: 

1. Add parameters to a URL
You can add parameters to a URL:

	<a href="welcome.asp?fname=John&lname=Smith">

Go to Welcome Page</a>


And retrieve the values in the "welcome.asp" file like this:

	<%

fname=Request.querystring("fname")

lname=Request.querystring("lname")

response.write("<p>Hello " & fname & " " & lname & "!</p>")

response.write("<p>Welcome to my Web site!</p>")%>


2. Use a form
You can use a form. The form passes the user input to "welcome.asp" when the user clicks on the Submit button: 

	<form method="post" action="welcome.asp">

First Name:  <input type="text" name="fname" value="">

Last Name: <input type="text" name="lname" value="">

<input type="submit" value="Submit">

</form>


Retrieve the values in the "welcome.asp" file like this:

	<%

fname=Request.form("fname")

lname=Request.form("lname")

response.write("<p>Hello " & fname & " " & lname & "!</p>")

response.write("<p>Welcome to my Web site!</p>")

%>


ASP Session Object

The Session object is used to store information about, or change settings for a user session. Variables stored in the Session object hold information about one single user, and are available to all pages in one application. 



The Session object 

When you are working with an application, you open it, do some changes and then you close it. This is much like a Session. The computer knows who you are. It knows when you start the application and when you end. But on the internet there is one problem: the web server does not know who you are and what you do because the HTTP address doesn't maintain state. 

ASP solves this problem by creating a unique cookie for each user. The cookie is sent to the client and it contains information that identifies the user. This interface is called the Session object. 

The Session object is used to store information about, or change settings for a user session. Variables stored in the Session object hold information about one single user, and are available to all pages in one application. Common information stored in session variables are name, id, and preferences. The server creates a new Session object for each new user, and destroys the Session object when the session expires. 



When does a Session Start?

A session starts when: 

· A new user requests an ASP file, and the Global.asa file includes a Session_OnStart procedure 

· A value is stored in a Session variable 

· A user requests an ASP file, and the Global.asa file uses the <object> tag to instantiate an object with session scope 



When does a Session End?

A session ends if a user has not requested or refreshed a page in the application for a specified period. By default, this is 20 minutes. 

If you want to set a timeout interval that is shorter or longer than the default, you can set the Timeout property.

The example below sets a timeout interval of 5 minutes:

	<%

Session.Timeout=5

%>


To end a session immediately, you may use the Abandon method:

	<%

Session.Abandon

%>


Note: The main problem with sessions is WHEN they should end. We do not know if the user's last request was the final one or not. So we do not know how long we should keep the session "alive". Waiting too long uses up resources on the server. But if the session is deleted too fast you risk that the user is coming back and the server has deleted all the information, so the user has to start all over again. Finding the right timeout interval can be difficult. 

Tip: If you are using session variables, store SMALL amounts of data in them. 



Store and Retrieve Session Variables 

The most important thing about the Session object is that you can store variables in it.

The example below will set the Session variable username to "Donald Duck" and the Session variable age to "50":

	<%

Session("username")="Donald Duck"

Session("age")=50

%>


When the value is stored in a session variable it can be reached from ANY page in the ASP application:

	Welcome <%Response.Write(Session("username"))%>


The line above returns: "Welcome Donald Duck".

You can also store user preferences in the Session object, and then access that preference to choose what page to return to the user.  

The example below specifies a text-only version of the page if the user has a low screen resolution: 

	<%If Session("screenres")="low" Then%> 

  This is the text version of the page

<%Else%> 

  This is the multimedia version of the page

<%End If%>




Remove Session Variables 

The Contents collection contains all session variables.

It is possible to remove a session variable with the Remove method.

The example below removes the session variable "sale" if the value of the session variable "age" is lower than 18: 

	<%

If Session.Contents("age")<18 then 

  Session.Contents.Remove("sale")

End If 

%>


To remove all variables in a session, use the RemoveAll method: 

	<%

Session.Contents.RemoveAll()

%>




Loop Through the Contents Collection

The Contents collection contains all session variables. You can loop through the Contents collection, to see what's stored in it: 

	<%

Session("username")="Donald Duck"

Session("age")=50

dim i

For Each i in Session.Contents

  Response.Write(i & "<br />")

Next

%>


Result:

	username

age


If you do not know the number of items in the Contents collection, you can use the Count property:

	<%

dim i

dim j

j=Session.Contents.Count

Response.Write("Session variables: " & j)

For i=1 to j

  Response.Write(Session.Contents(i) & "<br />")

Next

%>


Result:

	Session variables: 2

Donald Duck

50




Loop Through the StaticObjects Collection

You can loop through the StaticObjects collection, to see the values of all objects stored in the Session object:

	<%

dim i

For Each i in Session.StaticObjects

  Response.Write(i & "<br />")

Next

%>


ASP Application Object

A group of ASP files that work together to perform some purpose is called an application. The Application object in ASP is used to tie these files together.



Application Object

An application on the Web may be a group of ASP files. The ASP files work together to perform some purpose. The Application object in ASP is used to tie these files together.

The Application object is used to store and access variables from any page, just like the Session object. The difference is that ALL users share one Application object, while with Sessions there is one Session object for EACH user.

The Application object should hold information that will be used by many pages in the application (like database connection information). This means that you can access the information from any page. It also means that you can change the information in one place and the changes will automatically be reflected on all pages.



Store and Retrieve Application Variables

Application variables can be accessed and changed by any page in the application.

You can create Application variables in "Global.asa" like this:

	<script language="vbscript" runat="server">

Sub Application_OnStart

application("vartime")=""

application("users")=1

End Sub

</script>


In the example above we have created two Application variables: "vartime" and "users".

You can access the value of an Application variable like this:

	There are 

<%

Response.Write(Application("users"))

%> 

active connections.




Loop Through the Contents Collection

The Contents collection contains all application variables. You can loop through the Contents collection, to see what's stored in it:

	<%

dim i

For Each i in Application.Contents

  Response.Write(i & "<br />")

Next

%>


If you do not know the number of items in the Contents collection, you can use the Count property:

	<%

dim i

dim j

j=Application.Contents.Count

For i=1 to j

  Response.Write(Application.Contents(i) & "<br />")

Next

%>




Loop Through the StaticObjects Collection

You can loop through the StaticObjects collection, to see the values of all objects stored in the Application object:

	<%

dim i

For Each i in Application.StaticObjects

  Response.Write(i & "<br />")

Next

%>




Lock and Unlock

You can lock an application with the "Lock" method. When an application is locked, the users cannot change the Application variables (other than the one currently accessing it). You can unlock an application with the "Unlock" method. This method removes the lock from the Application variable:

	<%

Application.Lock

  'do some application object operations

Application.Unlock

%>


ASP Including Files

The #include directive is used to create functions, headers, footers, or elements that will be reused on multiple pages.



The #include Directive

You can insert the content of one ASP file into another ASP file before the server executes it, with the #include directive. The #include directive is used to create functions, headers, footers, or elements that will be reused on multiple pages.



How to Use the #include Directive

Here is a file called "mypage.asp":

	<html> 

<body> 

<h3>Words of Wisdom:</h3>

<p><!--#include file="wisdom.inc"--></p> 

<h3>The time is:</h3>

<p><!--#include file="time.inc"--></p>

</body> 

</html> 


Here is the "wisdom.inc" file:

	"One should never increase, beyond what is necessary,

the number of entities required to explain anything."


Here is the "time.inc" file:

	<%

Response.Write(Time)

%>


If you look at the source code in a browser, it will look something like this: 

	<html>

<body>

<h3>Words of Wisdom:</h3>

<p>"One should never increase, beyond what is necessary,

the number of entities required to explain anything."</p>

<h3>The time is:</h3>

<p>11:33:42 AM</p>

</body>

</html>




Syntax for Including Files

To include a file in an ASP page, place the #include directive inside comment tags:

	<!--#include virtual="somefilename"-->

or

<!--#include file ="somefilename"-->


The Virtual Keyword
Use the virtual keyword to indicate a path beginning with a virtual directory.

If a file named "header.inc" resides in a virtual directory named /html, the following line would insert the contents of "header.inc":

	<!-- #include virtual ="/html/header.inc" -->


The File Keyword
Use the file keyword to indicate a relative path. A relative path begins with the directory that contains the including file.

If you have a file in the html directory, and the file "header.inc" resides in html\headers, the following line would insert "header.inc" in your file:

	<!-- #include file ="headers\header.inc" -->


Note that the path to the included file (headers\header.inc) is relative to the including file. If the file containing this #include statement is not in the html directory, the statement will not work.

You can also use the file keyword with the syntax (..\) to include a file from a higher-level directory.



Tips and Notes

In the sections above we have used the file extension ".inc" for included files. Notice that if a user tries to browse an INC file directly, its content will be displayed. If your included file contains confidential information or information you do not want any users to see, it is better to use an ASP extension. The source code in an ASP file will not be visible after the interpretation. An included file can also include other files, and one ASP file can include the same file more than once. 

Important: Included files are processed and inserted before the scripts are executed.

The following script will not work because ASP executes the #include directive before it assigns a value to the variable:

	<%

fname="header.inc"

%>

<!--#include file="<%=fname%>"-->


You cannot open or close a script delimiter in an INC file. This script will not work:

	<%

For i = 1 To n

  <!--#include file="count.inc"-->

Next

%>


But this script will work:

	<% For i = 1 to n %>

<!--#include file="count.inc" -->

<% Next %>


ASP The Global.asa file

The Global.asa file is an optional file that can contain declarations of objects, variables, and methods that can be accessed by every page in an ASP application.



The Global.asa file

The Global.asa file is an optional file that can contain declarations of objects, variables, and methods that can be accessed by every page in an ASP application. All valid browser scripts (JavaScript, VBScript, JScript, PerlScript, etc.) can be used within Global.asa.

The Global.asa file can contain only the following:

· Application events 

· Session events 

· <object> declarations  

· TypeLibrary declarations 

· the #include directive 

Note: The Global.asa file must be stored in the root directory of the ASP application, and each application can only have one Global.asa file.



Events in Global.asa

In Global.asa you can tell the application and session objects what to do when the application/session starts and what to do when the application/session ends. The code for this is placed in event handlers. The Global.asa file can contain four types of events:

Application_OnStart - This event occurs when the FIRST user calls the first page from an ASP application. This event occurs after the Web server is restarted or after the Global.asa file is edited. The "Session_OnStart" event occurs immediately after this event. 

Session_OnStart - This event occurs EVERY time a NEW user requests his or hers first page in the ASP application.

Session_OnEnd - This event occurs EVERY time a user ends a session. A user ends a session after a page has not been requested by the user for a specified time (by default this is 20 minutes).

Application_OnEnd - This event occurs after the LAST user has ended the session. Typically, this event occurs when a Web server stops. This procedure is used to clean up settings after the Application stops, like delete records or write information to text files.

A Global.asa file could look something like this:

	<script language="vbscript" runat="server">

sub Application_OnStart

''''some code

end sub

sub Application_OnEnd

''''some code

end sub

sub Session_OnStart

''''some code

end sub

sub Session_OnEnd

''''some code

end sub

</script>


Note: We cannot use the ASP script delimiters (<% and %>) to insert scripts in the Global.asa file, we will have to put the subroutines inside the HTML <script> tag. 



<object> Declarations 

It is possible to create objects with session or application scope in Global.asa by using the <object> tag. 

Note: The <object> tag should be outside the <script> tag!

Syntax 
	<object runat="server" scope="scope" id="id"

{progid="progID"|classid="classID"}>

....

</object>


	Parameter
	Description

	scope 
	Sets the scope of the object (either Session or Application) 

	id 
	Specifies a unique id for the object 

	ProgID 
	An id associated with a class id. The format for ProgID is [Vendor.]Component[.Version] 

Either ProgID or ClassID must be specified.

	ClassID 
	Specifies a unique id for a COM class object. 

Either ProgID or ClassID must be specified.


Examples 
The first example creates an object of session scope named "MyAd" by using the ProgID parameter:

	<object runat="server" scope="session" id="MyAd"

progid="MSWC.AdRotator"></object>


The second example creates an object of application scope named "MyConnection" by using the ClassID parameter:

	<object runat="server" scope="application" id="MyConnection"

classid="Clsid:8AD3067A-B3FC-11CF-A560-00A0C9081C21">

</object>


The objects declared in the Global.asa file can be used by any script in the application:

	GLOBAL.ASA:

<object runat="server" scope="session" id="MyAd"

progid="MSWC.AdRotator">

</object>

You could reference the object "MyAd" from any page in the ASP application:

SOME .ASP FILE:

<%=MyAd.GetAdvertisement("/banners/adrot.txt")%> 




TypeLibrary Declarations

A TypeLibrary is a container for the contents of a DLL file corresponding to a COM object. By including a call to the TypeLibrary in the Global.asa file, the constants of the COM object can be accessed, and errors can be better reported by the ASP code. If your Web application relies on COM objects that have declared data types in type libraries, you can declare the type libraries in Global.asa.

Syntax 
	<!--METADATA TYPE="TypeLib"

file="filename"

uuid="typelibraryuuid"

version="versionnumber"

lcid="localeid"

-->


	Parameter
	Description

	file
	Specifies an absolute path to a type library. 

Either the file parameter or the uuid parameter is required

	uuid
	Specifies a unique identifier for the type library. 

Either the file parameter or the uuid parameter is required

	version
	Optional. Used for selecting version. If the requested version is not found, then the most recent version is used

	localeid
	Optional. The locale identifier to be used for the type library


Error Values
The server can return one of the following error messages:

	Error Code
	Description

	ASP 0222
	Invalid type library specification 

	ASP 0223
	Type library not found 

	ASP 0224
	Type library cannot be loaded

	ASP 0225
	Type library cannot be wrapped


Note: METADATA tags can appear anywhere in the Global.asa file (both inside and outside <script> tags). However, it is recommended that METADATA tags appear near the top of the Global.asa file. 



Restrictions

Restrictions on what you can include in the Global.asa file: 

· You can not display text that is written in the Global.asa file. This file can't display information 

· You can only use Server and Application objects in the Application_OnStart and Application_OnEnd subroutines. In the Session_OnEnd subroutine, you can use Server, Application, and Session objects. In the Session_OnStart subroutine you can use any built-in object 



How to use the Subroutines

Global.asa is often used to initialize variables. 

The example below shows how to detect the exact time a visitor first arrives on a Web site. The time is stored in a Session variable named "started", and the value of the "started" variable can be accessed from any ASP page in the application:

	<script language="vbscript" runat="server">

sub Session_OnStart

Session("started")=now()

end sub

</script>


Global.asa can also be used to control page access. 

The example below shows how to redirect every new visitor to another page, in this case to a page called "newpage.asp": 

	<script language="vbscript" runat="server">

sub Session_OnStart

Response.Redirect("newpage.asp")

end sub

</script>


And you can include functions in the Global.asa file. 

In the example below the Application_OnStart subroutine occurs when the Web server starts. Then the Application_OnStart subroutine calls another subroutine named "getcustomers". The "getcustomers" subroutine opens a database and retrieves a record set from the "customers" table. The record set is assigned to an array, where it can be accessed from any ASP page without querying the database: 

	<script language="vbscript" runat="server">

sub Application_OnStart

getcustomers

end sub

sub getcustomers 

set conn=Server.CreateObject("ADODB.Connection")

conn.Provider="Microsoft.Jet.OLEDB.4.0"

conn.Open "c:/webdata/northwind.mdb"

set rs=conn.execute("select name from customers")

Application("customers")=rs.GetRows

rs.Close

conn.Close

end sub

</script>




Global.asa Example 

In this example we will create a Global.asa file that counts the number of current visitors. 

· The Application_OnStart sets the Application variable "visitors" to 0 when the server starts 

· The Session_OnStart subroutine adds one to the variable "visitors" every time a new visitor arrives 

· The Session_OnEnd subroutine subtracts one from "visitors" each time this subroutine is triggered 

The Global.asa file: 

	<script language="vbscript" runat="server">

Sub Application_OnStart

Application("visitors")=0

End Sub

Sub Session_OnStart

Application.Lock

Application("visitors")=Application("visitors")+1

Application.UnLock

End Sub

Sub Session_OnEnd

Application.Lock

Application("visitors")=Application("visitors")-1

Application.UnLock

End Sub

</script>


To display the number of current visitors in an ASP file: 

	<html>

<head>

</head>

<body>

<p>

There are <%response.write(Application("visitors"))%>

online now!

</p>

</body>

</html>


ASP Response Object
The ASP Response object is used to send output to the user from the server.



Examples

Write text with ASP
This example demonstrates how to write text with ASP. 

<html>
<body>

<%
response.write("Hello World!")
%>

</body>
</html>
Format text with HTML tags in ASP
This example demonstrates how to combine text and HTML tags with ASP. 

<html>
<body>
<%
response.write("<h2>You can use HTML tags to format the text!</h2>")
%>

<%
response.write("<p style='color:#0000ff'>This text is styled with the style attribute!</p>")
%>
</body>
</html>
Redirect the user to a different URL
This example demonstrates how to redirect the user to a different URL. 

<%
if Request.Form("select")<>"" then
       Response.Redirect(Request.Form("select"))
end if
%>   

<html>
<body>

<form action="demo_redirect.asp" method="post">

<input type="radio" name="select" 
value="demo_server.asp">
Server Example<br>

<input type="radio" name="select" 
value="demo_text.asp">
Text Example<br><br>
<input type="submit" value="Go!">

</form></body></html>
Show a random link
This example demonstrates a link, each time you load the page, it will display one of two links: W3Schools.com! OR Refsnesdata.no! There is a 50% chance for each of them. 

<html>
<body>

<%
randomize()
r=rnd()
if r>0.5 then
  response.write("<a href='http://www.w3schools.com'>W3Schools.com!</a>")
else
  response.write("<a href='http://www.refsnesdata.no'>Refsnesdata.no!</a>")
end if
%>

<p>
This example demonstrates a link, each time you load the page, it will display 
one of two links: W3Schools.com! OR Refsnesdata.no! There is a 50% chance for 
each of them.
</p>

</body>
</html>
Controlling the buffer
This example demonstrates how you can control the buffer. 

<%
Response.Buffer=true
%>
<html>
<body>
<p>
This text will be sent to your browser when my response buffer is flushed.
</p>
<%
Response.Flush
%>
</body>
</html>
Clear the buffer
This example demonstrates how you can clear the buffer. 

<%
Response.Buffer=true
%>
<html>
<body>
<p>This is some text I want to send to the user.</p>
<p>No, I changed my mind. I want to clear the text.</p>
<%
Response.Clear
%>
</body>
</html>
End a script in the middle of processing and return the result
This example demonstrates how to end a script in the middle of processing. 

<html>
<body>
<p>I am writing some text. This text will never be<br>
<%
Response.End
%>
finished! It's too late to write more!</p>
</body>
</html>
Set how many minutes a page will be cached in a browser before it expires
This example demonstrates how to specify how many minutes a page will be cached in a browser before it expires. 

<%Response.Expires=-1%>
<html>
<body>
<p>This page will be refreshed with each access!</p>
</body>
</html>
Set a date/time when a page cached in a browser will expire
This example demonstrates how to specify a date/time a page cached in a browser will expire. 

<%
Response.ExpiresAbsolute=#May 05,2001 05:30:30#
%>
<html>
<body>
<p>This page will expire on May 05, 2001 05:30:30!</p>
</body>
</html>
Check if the user is still connected to the server
This example demonstrates how to check if a user is disconnected from the server. 

<html>
<body>

<%
If Response.IsClientConnected=true then
Response.Write("The user is still connected!")
else
Response.Write("The user is not connected!")
end if
%>

</body>
</html>
Set the type of content
This example demonstrates how to specify the type of content. 

<%
Response.ContentType="text/html"
%>
<html>
<body>

<p>This is some text</p>

</body>
</html>
Set the name of the character set
This example demonstrates how to specify the name of the character set. 

<%
Response.Charset="ISO8859-1"
%>
<html>
<body>

<p>This is some text</p>

</body>
</html>


Response Object

The ASP Response object is used to send output to the user from the server. Its collections, properties, and methods are described below:

Collections
	Collection
	Description

	Cookies
	Sets a cookie value. If the cookie does not exist, it will be created, and take the value that is specified


Properties
	Property
	Description

	Buffer
	Specifies whether to buffer the page output or not

	CacheControl
	Sets whether a proxy server can cache the output generated by ASP or not

	Charset
	Appends the name of a character-set to the content-type header in the Response object

	ContentType
	Sets the HTTP content type for the Response object

	Expires
	Sets how long (in minutes) a page will be cached on a browser before it expires

	ExpiresAbsolute
	Sets a date and time when a page cached on a browser will expire

	IsClientConnected
	Indicates if the client has disconnected from the server

	Pics
	Appends a value to the PICS label response header

	Status
	Specifies the value of the status line returned by the server


Methods
	Method
	Description

	AddHeader
	Adds a new HTTP header and a value to the HTTP response

	AppendToLog
	Adds a string to the end of the server log entry

	BinaryWrite
	Writes data directly to the output without any character conversion

	Clear
	Clears any buffered HTML output

	End
	Stops processing a script, and returns the current result

	Flush
	Sends buffered HTML output immediately

	Redirect
	Redirects the user to a different URL

	Write
	Writes a specified string to the output


ASP Request Object

The ASP Request object is used to get information from the user.



QueryString Collection Examples

Send query information when a user clicks on a link
This example demonstrates how to send some extra query information to a page within a link, and retrieve that information on the destination page (which is, in this example, the same page).

<html>
<body>

<a href="demo_simplequerystring.asp?color=green">Example</a>

<%
Response.Write(Request.QueryString)
%>

</body>
</html>
A QueryString collection in its simplest use
This example demonstrates how the QueryString collection retrieves the values from a form. The form uses the GET method, which means that the information sent is visible to everybody (in the address field). The GET method also limits the amount of information that can be sent.

<html>
<body>

<form action="demo_simplereqquery.asp" method="get">
First name: <input type="text" name="fname"><br>
Last name: <input type="text" name="lname"><br>
<input type="submit" value="Submit">
</form>

<%
Response.Write(Request.QueryString)
%>

</body>
</html>
How to use information from forms
This example demonstrates how to use the values retrieved from a form. We use the QueryString collection. The form uses the get method.

<html>
<body>
<form action="demo_reqquery.asp" method="get">
Your name: <input type="text" name="fname" size="20">
<input type="submit" value="Submit">
</form>
<%
dim fname
fname=Request.QueryString("fname")
If fname<>"" Then
      Response.Write("Hello " & fname & "!<br />")
      Response.Write("How are you today?")
End If
%>
</body>
</html>
More information from a form
This example demonstrates what the QueryString contains if several input fields have the same name. It shows how to separate input fields with equal names from each other. It also shows how to use the Count keyword to count the "name" property. The form uses the get method.

<html>
<body>

<%
If Request.QueryString<>"" Then
      If Request.QueryString("name")<>", " Then
           name1=Request.QueryString("name")(1)
           name2=Request.QueryString("name")(2)
      end if
end if
%>

<form action="demo_reqquery2.asp" method="get">
First name:
<input type="text" name="name" value="<%=name1%>">
<br>
Last name:
<input type="text" name="name" value="<%=name2%>">
<br>
<input type="submit" value="Submit">
</form>
<hr>
<%
If Request.QueryString<>"" Then
      Response.Write("<p>")
      Response.Write("The information received from the form was:")
      Response.Write("</p><p>")
      Response.Write("name=" & Request.QueryString("name"))
      Response.Write("</p><p>")
      Response.Write("The name property's count is: ")
      Response.Write(Request.QueryString("name").Count)
      Response.Write("</p><p>")
      Response.Write("First name=" & name1)
      Response.Write("</p><p>")
      Response.Write("Last name=" & name2)
      Response.Write("</p>")
end if%></body></html>
Form Collection Examples

A form collection in its simplest use
This example demonstrates how the Form collection retrieves the values from a form. The form uses the POST method, which means that the information sent is invisible to others, and it has no limits (you can send a large amount of information).

How to use information from forms
This example demonstrates how to use the values retrieved from a form. We use the Form collection. The form uses the post method.

More information from a form
This example demonstrates what the Form collection contains if several input fields have the same name. It shows how to separate input fields with equal names from each other. It also shows how to use the Count keyword to count the "name" property. The form uses the post method.

A form with radio buttons
This example demonstrates how to interact with the user through radio buttons, with the Form collection. The form uses the post method.

A form with checkboxes
This example demonstrates how to interact with the user through checkboxes, with the Form collection. The form uses the post method.

Other Examples

Get the server variables
This example demonstrates how to find out the visitors (yours) browser type, IP address, and more with the ServerVariables collection. 

Create a welcome cookie
This example demonstrates how to create a Welcome Cookie with the Cookies Collection.

Find the total number of bytes the user sent
This example demonstrates how to use the TotalBytes property to find out the total number of bytes the user sent in the Request object.



Request Object

When a browser asks for a page from a server, it is called a request. The ASP Request object is used to get information from the user. Its collections, properties, and methods are described below:

Collections
	Collection
	Description

	ClientCertificate
	Contains all the field values stored in the client certificate

	Cookies
	Contains all the cookie values sent in a HTTP request

	Form
	Contains all the form (input) values from a form that uses the post method

	QueryString
	Contains all the variable values in a HTTP query string

	ServerVariables
	Contains all the server variable values


Properties
	Property
	Description

	TotalBytes
	Returns the total number of bytes the client sent in the body of the request


Methods
	Method
	Description

	BinaryRead
	Retrieves the data sent to the server from the client as part of a post request and stores it in a safe array


